Examples with gfun[istranscendental]

Including most of the examples mentioned in
``On deciding transcendence of power series''

Alin Bostan, Bruno Salvy, Michael Singer

April 2025

```
> restart;
```

Start timer:

```
> T0:=time():
```

The version of gfun should be at least 4.05

```
> gfun:-version();
```

4.10

Load the function

```
> with(gfun,istranscendental);
```

$[istranscendental]$

This displays more information on intermediate computations:

```
> infolevel[istranscendental]:=3:
```

External data used in some of the examples:

```
> read `data-ex-istrans.m`;
```

Simple examples

Exponential:

```
> istranscendental({diff(y(z),z)-y(z),y(0)=1},y(z));
```

true, "irregular singularity at infinity"

For simple functions, use gfun to get the differential equation by itself:

```
> f:=log(1-z);
```

$$f := \ln(1-z)$$

```
> deq:=gfun:-holexprtodiffeq(f,y(z));
```

$$deq := \left\{\left(\frac{d}{dz}\, y(z)\right)(-1+z) - 1,\, y(0) = 0\right\}$$

```
> istranscendental(deq,y(z));
```

true, "multiple root of multiplicity 2 of the indicial equation ==> ln at 1"

If the equation is not minimal, istranscendental will minimize it first:

```
> f:=(1-z)*log((1-z))+1/(1-z);
```

$$f := (1-z)\ln(1-z) + \frac{1}{1-z}$$

```
> deq:=gfun:-holexprtodiffeq(f,y(z));
```

$$deq := \left\{ (z^2 - 2z + 1)\left(\frac{d^2}{dz^2}y(z)\right) + \left(\frac{d}{dz}y(z)\right)(-1+z) - y(z) + 2z - 2, y(0) = 1, \mathrm{D}(y)(0) = 0 \right\}$$

```
> istranscendental(deq,y(z));
istranscendental: input diff eqn of order 3, minimal one of order 2
```

true, "logarithmic singularity at 1"

Apéry sequence (Ex. 4, sec. 2.2)

Guess the differential equation from the first terms:

```
> L:=[seq(add(binomial(n,k)^2*binomial(n+k,k)^2,k=0..n),n=0..30)]:
> deq:=gfun:-listtodiffeq(L,y(z),[ogf])[1];
```

$$deq := \left\{ (z-5)y(z) + (7z^2 - 112z + 1)\left(\frac{d}{dz}y(z)\right) + (6z^3 - 153z^2 + 3z)\left(\frac{d^2}{dz^2}y(z)\right) + (z^4 - 34z^3 + z^2)\left(\frac{d^3}{dz^3}y(z)\right), y(0) = 1, \mathrm{D}(y)(0) = 5, \mathrm{D}^{(2)}(y)(0) = 146 \right\}$$

This is the well-known differential equation and it lets one prove that the generating function is transcendental:

```
> istranscendental(deq,y(z));
```

true, "multiple root of multiplicity 3 of the indicial equation ==> ln at 0"

Walks with small steps in the quarter plane (Ex. 5, sec. 2.2)

The operators are available at https://specfun.inria.fr/chyzak/ssw/ct-P.mpl

```
> T0:=time():nn:=0:
> for i to 19 do for a in [0,1] do for b in [0,1] do for c in ["xy"]
  do
     ind:=[i,a,b,c];
     if not assigned(P[i,a,b,c]) or not assigned(Qt[i,a,b]) then next
  fi;
     dop:=P[op(ind)];
     deq:=DEtools[diffop2de](dop,[dt,t],y(t));
     indeq:=op(1,DEtools[indicialeq](deq,t,0,y(t)));
     ord:=max(select(type,map2(op,1,roots(indeq,t)),integer));
     for j from ord+1 to 100 do
         ini:=series(Qt[op(1..-2,ind)],t,j);
         if order(ini)>ord then break fi
```

```
    od;
    deq:={deq,seq((D@@i)(y)(0)=coeff(ini,t,i)*i!,i=0..ord)};
    st:=time();
    res:=[istranscendental(deq,y(t))];
    if res[1]<>FAIL then nn:=nn+1;
           print(ind,res[2],time()-st)
    else
       print(ind,res[1],time()-st,Qt[op(1..-2,ind)])
    fi
  od od od od:nn,time()-T0;
```

[1, 0, 0, "xy"], "logarithmic singularity at 0", 0.219

istranscendental: input diff eqn of order 4, minimal one of order 3

[1, 0, 1, "xy"], "logarithmic singularity at 0", 0.173

istranscendental: input diff eqn of order 4, minimal one of order 3

[1, 1, 0, "xy"], "logarithmic singularity at 0", 0.098

[1, 1, 1, "xy"], "logarithmic singularity at 0", 0.239

[2, 0, 0, "xy"], "multiple root of multiplicity 2 of the indicial equation ==> ln at 0", 0.052

istranscendental: input diff eqn of order 4, minimal one of order 2

[2, 0, 1, "xy"], "logarithmic singularity at 0", 0.114

istranscendental: input diff eqn of order 4, minimal one of order 2

[2, 1, 0, "xy"], "logarithmic singularity at 0", 0.216

[2, 1, 1, "xy"], "multiple root of multiplicity 2 of the indicial equation ==> ln at 0", 0.049

[3, 0, 0, "xy"], "multiple root of multiplicity 2 of the indicial equation ==> ln at 0", 0.115

[3, 0, 1, "xy"], "multiple root of multiplicity 2 of the indicial equation ==> ln at 0", 0.139

[3, 1, 0, "xy"], "multiple root of multiplicity 2 of the indicial equation ==> ln at 0", 0.101

[3, 1, 1, "xy"], "multiple root of multiplicity 2 of the indicial equation ==> ln at 0", 0.066

[4, 0, 0, "xy"], "multiple root of multiplicity 2 of the indicial equation ==> ln at 0", 0.089

[4, 0, 1, "xy"], "multiple root of multiplicity 2 of the indicial equation ==> ln at 0", 0.089

[4, 1, 0, "xy"], "multiple root of multiplicity 2 of the indicial equation ==> ln at 0", 0.089

[4, 1, 1, "xy"], "multiple root of multiplicity 2 of the indicial equation ==> ln at 0", 0.265

[5, 0, 0, "xy"], "logarithmic singularity at 0", 0.084

istranscendental: input diff eqn of order 6, minimal one of order 5

[5, 0, 1, "xy"], "multiple root of multiplicity 2 of the indicial equation ==> ln at 0", 0.723

[5, 1, 0, "xy"], "multiple root of multiplicity 2 of the indicial equation ==> ln at 0", 0.252

[5, 1, 1, "xy"], "multiple root of multiplicity 2 of the indicial equation ==> ln at 0", 0.327

[6, 0, 0, "xy"], "logarithmic singularity at 0", 0.175

[6, 0, 1, "xy"], "multiple root of multiplicity 2 of the indicial equation ==> ln at 0", 1.109

[6, 1, 0, "xy"], "multiple root of multiplicity 2 of the indicial equation ==> ln at 0", 0.130

[6, 1, 1, "xy"], "multiple root of multiplicity 2 of the indicial equation ==> ln at 0", 0.617

[7, 0, 0, "xy"], "logarithmic singularity at 0", 0.092

istranscendental: input diff eqn of order 6, minimal one of order 5

[7, 0, 1, "xy"], "multiple root of multiplicity 2 of the indicial equation ==> ln at 0", 0.689
[7, 1, 0, "xy"], "multiple root of multiplicity 2 of the indicial equation ==> ln at 0", 0.050
[7, 1, 1, "xy"], "multiple root of multiplicity 2 of the indicial equation ==> ln at 0", 0.539
[8, 0, 0, "xy"], "logarithmic singularity at 0", 0.171
[8, 0, 1, "xy"], "multiple root of multiplicity 2 of the indicial equation ==> ln at 0", 0.767
[8, 1, 0, "xy"], "multiple root of multiplicity 2 of the indicial equation ==> ln at 0", 0.120
[8, 1, 1, "xy"], "multiple root of multiplicity 2 of the indicial equation ==> ln at 0", 0.320
[9, 0, 0, "xy"], "multiple root of multiplicity 2 of the indicial equation ==> ln at 0", 0.119
[9, 0, 1, "xy"], "multiple root of multiplicity 2 of the indicial equation ==> ln at 0", 1.077
[9, 1, 0, "xy"], "multiple root of multiplicity 2 of the indicial equation ==> ln at 0", 0.351
[9, 1, 1, "xy"], "multiple root of multiplicity 2 of the indicial equation ==> ln at 0", 0.487
[10, 0, 0, "xy"], "multiple root of multiplicity 2 of the indicial equation ==> ln at 0", 0.139
[10, 0, 1, "xy"], "multiple root of multiplicity 2 of the indicial equation ==> ln at 0", 1.323
[10, 1, 0, "xy"], "multiple root of multiplicity 2 of the indicial equation ==> ln at 0", 0.128
[10, 1, 1, "xy"], "multiple root of multiplicity 2 of the indicial equation ==> ln at 0", 0.489
[11, 0, 0, "xy"], "logarithmic singularity at 0", 0.061

```
istranscendental: input diff eqn of order 5, minimal one of order 4
```

[11, 0, 1, "xy"], "logarithmic singularity at 0", 0.703
[11, 1, 0, "xy"], "multiple root of multiplicity 2 of the indicial equation ==> ln at 0", 0.055

```
istranscendental: input diff eqn of order 5, minimal one of order 4
```

[11, 1, 1, "xy"], "multiple root of multiplicity 2 of the indicial equation ==> ln at 0", 0.387
[12, 0, 0, "xy"], "logarithmic singularity at 0", 0.361

```
istranscendental: input diff eqn of order 6, minimal one of order 5
```

[12, 0, 1, "xy"], "multiple root of multiplicity 2 of the indicial equation ==> ln at 0", 0.734
[12, 1, 0, "xy"], "multiple root of multiplicity 2 of the indicial equation ==> ln at 0", 0.371
[12, 1, 1, "xy"], "multiple root of multiplicity 2 of the indicial equation ==> ln at 0", 0.310
[13, 0, 0, "xy"], "multiple root of multiplicity 2 of the indicial equation ==> ln at 0", 0.114
[13, 0, 1, "xy"], "multiple root of multiplicity 2 of the indicial equation ==> ln at 0", 1.052
[13, 1, 0, "xy"], "multiple root of multiplicity 2 of the indicial equation ==> ln at 0", 0.127
[13, 1, 1, "xy"], "multiple root of multiplicity 2 of the indicial equation ==> ln at 0", 0.746
[14, 0, 0, "xy"], "multiple root of multiplicity 2 of the indicial equation ==> ln at 0", 0.127
[14, 0, 1, "xy"], "multiple root of multiplicity 2 of the indicial equation ==> ln at 0", 1.053
[14, 1, 0, "xy"], "multiple root of multiplicity 2 of the indicial equation ==> ln at 0", 0.379
[14, 1, 1, "xy"], "multiple root of multiplicity 2 of the indicial equation ==> ln at 0", 0.481
[15, 0, 0, "xy"], "logarithmic singularity at 0", 0.062

```
istranscendental: input diff eqn of order 5, minimal one of order 4
```

[15, 0, 1, "xy"], "logarithmic singularity at 0", 0.777
[15, 1, 0, "xy"], "multiple root of multiplicity 2 of the indicial equation ==> ln at 0", 0.051

```
istranscendental: input diff eqn of order 5, minimal one of order 4
```

[15, 1, 1, "xy"], "multiple root of multiplicity 2 of the indicial equation ==> ln at 0", 0.447

[16, 0, 0, "xy"], "logarithmic singularity at 0", 0.119

```
istranscendental: input diff eqn of order 6, minimal one of order 5
```

[16, 0, 1, "xy"], "multiple root of multiplicity 2 of the indicial equation ==> ln at 0", 1.111

[16, 1, 0, "xy"], "multiple root of multiplicity 2 of the indicial equation ==> ln at 0", 0.368

[16, 1, 1, "xy"], "multiple root of multiplicity 2 of the indicial equation ==> ln at 0", 0.354

[17, 0, 0, "xy"], "logarithmic singularity at 0", 0.095

```
istranscendental: input diff eqn of order 6, minimal one of order 4
```

[17, 0, 1, "xy"], "logarithmic singularity at 0", 0.714

```
istranscendental: input diff eqn of order 6, minimal one of order 4
```

[17, 1, 0, "xy"], "logarithmic singularity at 0", 0.456

```
istranscendental: input diff eqn of order 3, minimal one of order 2
```

$$[17, 1, 1, \text{"xy"}], \mathit{FAIL}, 0.130, \frac{1 - t - \sqrt{-3\,t^2 - 2\,t + 1}}{2\,t^2}$$

```
istranscendental: input diff eqn of order 4, minimal one of order 3
```

[18, 0, 0, "xy"], "logarithmic singularity at 0", 0.216

```
istranscendental: input diff eqn of order 4, minimal one of order 2
```

$$[18, 0, 1, \text{"xy"}], \mathit{FAIL}, 0.153, \frac{(1 - 6\,t)\sqrt{-12\,t^2 - 4\,t + 1} - 4\,t^2 + 8\,t - 1}{32\,t^3}$$

```
istranscendental: input diff eqn of order 4, minimal one of order 2
```

$$[18, 1, 0, \text{"xy"}], \mathit{FAIL}, 0.324, \frac{(1 - 6\,t)\sqrt{-12\,t^2 - 4\,t + 1} - 4\,t^2 + 8\,t - 1}{32\,t^3}$$

```
istranscendental: input diff eqn of order 3, minimal one of order 2
```

$$[18, 1, 1, \text{"xy"}], \mathit{FAIL}, 0.139, \frac{1 - 2\,t - \sqrt{-12\,t^2 - 4\,t + 1}}{8\,t^2}$$

```
istranscendental: input diff eqn of order 4, minimal one of order 3
```

[19, 0, 0, "xy"], "logarithmic singularity at 0", 0.173

[19, 0, 1, "xy"], "logarithmic singularity at 0", 0.049

```
istranscendental: input diff eqn of order 4, minimal one of order 3
```

[19, 1, 0, "xy"], "logarithmic singularity at 0", 0.186

```
istranscendental: input diff eqn of order 4, minimal one of order 3
```

[19, 1, 1, "xy"], "logarithmic singularity at 0", 0.105

72, 26.383

```
> unassign('a','b','c');
```

Tricky hypergeometric series (special case of Ex. 9, sec. 2.5)

```
> deq:=gfun:-holexprtodiffeq(hypergeom([1/6,5/6],[7/6],z),y(z));
```

$$deq := \left\{ (36z^2 - 36z)\left(\frac{d^2}{dz^2}y(z)\right) + (72z-42)\left(\frac{d}{dz}y(z)\right) + 5y(z), y(0)=1 \right\}$$

```
> istranscendental(deq,y(z));
```

FAIL

But this is transcendental as can be seen by applying the criterion of Fürnsinn & Yurkevich.

Powers of binomial(2n,n) (Stanley 1980 sec. 4. (g))

```
> for i to 8 do i,istranscendental(gfun:-listtodiffeq([seq(binomial
  (2*n,n)^i,n=0..100)],y(z),[ogf])[1],y(z)) od;
```

1, *FAIL*

2, *true*, "multiple root of multiplicity 2 of the indicial equation ==> ln at 0"

3, *true*, "multiple root of multiplicity 3 of the indicial equation ==> ln at 0"

4, *true*, "multiple root of multiplicity 4 of the indicial equation ==> ln at 0"

5, *true*, "multiple root of multiplicity 5 of the indicial equation ==> ln at 0"

6, *true*, "multiple root of multiplicity 6 of the indicial equation ==> ln at 0"

7, *true*, "multiple root of multiplicity 7 of the indicial equation ==> ln at 0"

8, *true*, "multiple root of multiplicity 8 of the indicial equation ==> ln at 0"

The case i=1 is algebraic. The other ones are proved transcendental. The general case can be proved by the Beukers-Heckman criterion.

Example from 3-dimensional lattice walks confined to the positive octant

The differential equation of order 11 with coefficients of degree 73 from Bostan-Bousquet-Mélou-Kauers-Melczer 2016

```
> st:=time():istranscendental(deqBBMKM16,y(t));time()-st;
```

true, "multiple root of multiplicity 2 of the indicial equation ==> ln at 0"

28.442

Lattice Green functions of the face-centered cubic lattice (sec. 5.1)

The operators are available at http://www.koutschan.de/data/fcc1/

We also compute the initial conditions that are needed to test transcendence

```
> t:=proc(d,n,j) option remember;
  local i,p,q; if d=2 then add(binomial(2*p,p)*binomial(2*j,2*p-n)*
  binomial(2*n+2*j-2*p,n+j-p),p=iquo(n+1,2)..iquo(n+2*j,2))
  else add(add(binomial(n,p)*binomial(2*j,2*q+p-n)*binomial(2*n+2*
  j-2*p-2*q,n+j-p-q)*t(d-1,p,q),q=iquo(n-p+1,2)..iquo(n-p+2*j,2)),p=
  0..n) fi end:
> ini:=proc(d,n) t(d,n,0)/4^n/binomial(d,2)^n end:
```

```
> for K from 3 to 7 do
      dop:=fcc[K];
      deq:=DEtools[diffop2de](dop,[Dx,x],y(x));
      ord:=degree(dop,Dx);
      iniv:=seq((D@@i)(y)(0)=ini(K,i)*i!,i=0..ord-1);
      st:=time();
      printf("dim = %d order = %d deg = %d. Transc.: %s (%g sec.)\n",
  K,ord,degree(dop,x),gfun:-istranscendental({deq,iniv},y(x))[2],time
  ()-st)
  od:
dim = 3 order = 3 deg = 5. Transc.: multiple root of multiplicity 3 of
the indicial equation ==> ln at 0 (0.041 sec.)
dim = 4 order = 4 deg = 10. Transc.: multiple root of multiplicity 4 of
the indicial equation ==> ln at 0 (0.116 sec.)
dim = 5 order = 6 deg = 17. Transc.: multiple root of multiplicity 5 of
the indicial equation ==> ln at 0 (0.746 sec.)
dim = 6 order = 8 deg = 43. Transc.: multiple root of multiplicity 6 of
the indicial equation ==> ln at 0 (4.699 sec.)
dim = 7 order = 11 deg = 68. Transc.: multiple root of multiplicity 7
of the indicial equation ==> ln at 0 (27.258 sec.)
```

Algebraic Diagonals (sec. 5.2)

```
> ratf := 1/(1 - x - x*y)/(1 - 5*x - 7*y*z - 13*z^2);
```

$$ratf := \frac{1}{(-x\,y-x+1)\,(-7\,y\,z-13\,z^2-5\,x+1)}$$

An operator annihilating the diagonal, computed by Koutschan's HolonomicFunctions:

```
> dop:=-4*x*(-1+140*x)*(256-34496*x-183689*x^2-1850338*x^3+4055917*
  x^4-512785\
  0*x^5+1482975*x^6)*(14623835337785344-16552972466349670 4*
  x+29577904419\
  757262848*x^2-28305468650021657364 48*x^3+38706974756219246802176*
  x^4-7\
  18962748243620748043764*x^5+95705014229635715512 43803*x^6
  -115366881062\
  736646674535330*x^7+30250494083543394461346185 0*
  x^8+184007806603040469\
  536669875*x^9+7507474803521988791324371 25*
  x^10+470046239499387650579 82\
  125*x^11+3610620941003083256207562 5*x^12
  -1417033807047494051401781 25*x\
  ^13+430987702605206449828125 0*x^14-609650350201196039531250*
  x^15+72616\
  552225877460937 5*x^16)*Dx^5-4*
  (-112311055394191441 92+7537057707163121 6\
  14848*x-860235096571406245691392*x^2+1340081961441326179752345 6*x^3
```

```
-21\
5776296492779667302309478 4*x^4+1111858778848147640704664238 08*x^5
-7223\
348520752268271354688881 92*x^6+24747028986872013904445069480788*x^7
-26\
156939078216398391407938133745 0*
x^8+35636642480978395730104187032431 25\
*x^9-257385105081605734014347913987392 15*
x^10+467386026087436823384988\
593411940200*x^11-2387463200242373165757458752745078125*
x^12+420626119\
0861299204874351062281311750*x^13
-57193070919658815390927430855494976 2\
5*x^14+5251278143338851268089648261048488750*x^15
-9270519365393900531 3\
5211904750027187 5*x^16+3663945008658483174233604533968681250*x^17
-1212\
1256899316604926725026851899843 75*
x^18+18750711848345136840917334665 68\
125000*x^19-669374750077762218050356608987890625*
x^20+2425951648938873\
6549149146224609375*x^21-241194877539336563962281626953125 0*
x^22+27137\
509947366998163714843750 00*x^23)*Dx^4+
(-391533059023333427773 44+905634\
2501166067610550272*x+261107333129513801723412 48*
x^2+1159821172788911 7\
746419220480*x^3-638451694556419791627548889088*x^4
-191995169105679740\
7762475602432*x^5-2366612853842666212200324374216 96*
x^6+25538536416282\
9142878049773959704 0*x^7-4538556757286960285404963256285454 5*
x^8+61641\
4393116578533615435918411045540*x^9
-8288930699592315414779377558656220\
225*x^10+4213218878641812338894368535453089587 5*x^11
-1051729330326509 3\
9052827939466389439500*
x^12+136559383775947622021382191396138332750*x^1\
3-951311939574781315630226273636777 08750*
x^14+21526916150218206251974 6\
263907072565625*x^15-1072256936541533745869189105377384562 50*
x^16+1459\
9514634985406915561649416989718750*x^17
-36897768499553945669540335229 2\
14375000*x^18+170333503781855703040898818703636718 75*x^19
-545242254170\
17397281226387875195312 5*x^20+56107899758065058395593646289062500*
```

```
x^21\
-593465514605150819410498535156 25*x^22)*Dx^3-3*
(-191232747342876533535\
5392-562190970711813447756O2176*x-17163599190964444273568645 12*
x^2+949\
41850581804886402678906 88*x^3+16456596687064138011906805 59104*
x^4+1395\
96398196177765927767520564608*x^5
-199138421460857189700069032938 5120*x\
^6+33447223893135051679514188341836380*x^7
-386966631796327401192372126\
023944105*x^8+355038022457232840036629865934255 1725*x^9
-20525167290259\
376725986365502540279500*
x^10+7295188957107642173684989933891 8312125*x\
^11-11033516295364856021927463627058 9638250*
x^12+547703731220536904560\
46754333005888125*x^13-1404593725740511666460866649891 67243750*
x^14+10\
60505600551177327417258788400605 71875*x^15
-227218452947370307572991178\
6257218750*x^16+197974778802696289406071991528542 96875*x^17
-1309006909\
85540057176865510569875 00000*
x^18+3658713802003688138628606198808 59375\
*x^19-38372920342620659099414231748046875*
x^20+376463152397424488929 55\
859375000*x^21)*Dx^2-30*(-168352792349337876588 1344
-107974267566978406\
726500352*x+75172483717189407731009945 6*x^2
-580845220028527667624673 68\
704*x^3+73510171938377616226923147 6608*x^4
-113235663612807246751333 33\
7380704*x^5+1677701139288149925349270563034790*x^6
-155786514861541961 1\
3405461292552840*x^7+10637334491154536996926791914 6628815*x^8
-85648502\
370320471350506707087733 7450*
x^9+434625406922979634697226616430 6948000\
*x^10-89247111846887822926999539846449 81875*
x^11+44050452304540184151 7\
747947690135 4125*x^12-70696776284326700671751892345217 31250*
x^13+10884\
89605171500158439935283736973 1250*
x^14+17831546821515067852684820572 49\
71875*x^15+818031856929418801272433583843 90625*x^16
-95646175869324151\
983252154572281250 0*x^17+2307124579690673501314908205546 8750*x^18
```

```
-2391\
441810930508818244764955078125*
x^19+2126329997522089074226875000000*x^2\
0)*Dx-150*(-27385406242070319773777792+22574949013749970032656384*
x-890\
1746185758430894921400032*x^2+770459328736943297024591316488*x^3
-1923668\
5437337518569154830127336*x^4+35095543341018969180787230555488*x^5
-3369\
25136512106935045563407770710*
x^6+2304155501884555688602247858325230*x\
^7-1568724023029985552330743177003625 5*
x^8+8369232334925760215186768 97\
80848475*x^9-248980587664770049523151681457085125*
x^10+257886135319996\
15796853013411005450 0*x^11-8088199189020296489711632689295125*
x^12+470\
05173200672154506539849926060437 5*
x^13+2386797274175025248308207936865\
625*x^14+6232133397004683330751718259243750*x^15
-26103076918923794027 8\
52077638890625*x^16+53610627133201217122990744382812 5*x^17
-51063149608\
162143672839973046875*x^18+39086948483861931511523437500*x^19):
> map2(degree,dop,[Dx,x]);
```

$$[5, 24]$$

Order 5 and degree 24.

It has logarithms at 0:

```
> DEtools[formal_sol](dop,[Dx,x],x=0,order=3);
```

$$\left[\frac{1}{6}+\left(-\frac{7322885597937\ln(x)}{446284037408}+\frac{10444105568561}{892568074816}\right)x+\left(-\frac{4459637329143633\ln(x)}{1785136149632}+\frac{1333149804448667428368909 55}{497923605112962858396 16}\right)x^2+\mathrm{O}(x^3),\ \left(\frac{\ln(x)}{2}+\frac{3}{4}\right)x+\left(\frac{609\ln(x)}{8}+\frac{58282944119463}{446284037408}\right)x^2+\mathrm{O}(x^3),\ \frac{1}{2}x+\frac{24309571771779}{223142018704}x^2+\mathrm{O}(x^3),\ -x^2+\mathrm{O}(x^3),\ \mathrm{O}(x^3)\right]$$

Compute 5 initial conditions:

```
> N:=5:
> S:=mtaylor(ratf,[x,y,z],3*N):
> diag:=add(coeff(coeff(coeff(S,x,i),y,i),z,i)*x^i,i=0..N);
```

$$diag := 117114971\,x^4+953729\,x^3+8147\,x^2+77\,x+1$$

```
> deq:={DEtools[diffop2de](dop,[Dx,x],y(x)),seq((D@@i)(y)(0)=coeff
  (diag,x,i)*i!,i=0..4)}:
> istranscendental(deq,y(x));
```

```
istranscendental: input diff eqn of order 5, minimal one of order 3
```

$$FAIL$$

Proof that a more general diagonal is algebraic and computation of an annihilating polynomial

```
> F:=1/(1-a*x-b*y*z-c*z^2)/(1-x-x*y);
```

$$F := \frac{1}{\left(-b\,y\,z - c\,z^2 - a\,x + 1\right)\left(-x\,y - x + 1\right)}$$

First, write the diagonal as a residue

```
> G:=normal(subs(z=t/x/y,F)/x/y);
```

$$G := \frac{x\,y}{\left(a\,x^3\,y^2 + b\,t\,x\,y^2 - x^2\,y^2 + c\,t^2\right)\left(x\,y + x - 1\right)}$$

Extract the factor of the denominator that has roots tending to 0 when t->0:

```
> pol:=select(has,denom(G),t);
```

$$pol := a\,x^3\,y^2 + b\,t\,x\,y^2 - x^2\,y^2 + c\,t^2$$

Computation of a polynomial having the residues at the roots of pol as roots:

```
> numer(G*pol-T*diff(pol,y)):
> resultant(%,pol,y);
```

$$x^2\,c\,t^2\left(4\,T^2\,a^3\,x^9 - 8\,T^2\,a^3\,x^8 + [\ldots 47\ terms\ldots] + b\,t\,x - x^2\right)$$

Sum of its roots (= sum of the residues):

```
> ff:=normal(-coeff(%,T,1)/coeff(%,T,2));
```

$$ff := \frac{x - 1}{a\,x^4 - 2\,a\,x^3 + b\,t\,x^2 + c\,t^2\,x + a\,x^2 - 2\,b\,t\,x - x^3 + b\,t + 2\,x^2 - x}$$

The denominator is irreducible:

```
> den:=factor(denom(ff));
```

$$den := a\,x^4 - 2\,a\,x^3 + b\,t\,x^2 + c\,t^2\,x + a\,x^2 - 2\,b\,t\,x - x^3 + b\,t + 2\,x^2 - x$$

Behaviour of the roots of the denominator as t->0:

```
> gfun:-algeqtoseries(den,t,x,3);
```

$$\left[\frac{1}{a} - b\,t - \frac{\left(b^2\,a^2 - 2\,b^2\,a + b^2 + c\right)a}{a^2 - 2\,a + 1}\,t^2 + \mathrm{O}\left(t^3\right),\ 1 + RootOf\left(\left(a - 1\right)\_Z^2 + c\right)t + \mathrm{O}\left(t^2\right),\ b\,t + b^2\,a\,t^2 + \left(2\,b^3\,a^2 + c\,b\right)t^3 + \mathrm{O}\left(t^4\right)\right]$$

Only one root tends to 0. The residue is a root of

```
> pol:=select(has,resultant(numer(ff)-T*diff(den,x),den,x),T);
```

$$pol := -27\,T^4\,a^2\,c^3\,t^6 + 180\,T^4\,a^2\,b\,c^2\,t^5 + [\ldots 49\ terms\ldots] - T + a$$

Check that this has the diagonal for root:

```
> gfun:-algeqtoseries(pol,t,T,5)[1]:
> map(normal,subs(RootOf((a - 1)*_Z^2 + _Z - a)=1,%));
```

$$1 + \left(2\,a + 1\right)b\,t + \left(6\,b^2\,a^2 + 3\,b^2\,a + b^2 + c\right)t^2 + b\left(20\,a^3\,b^2 + 10\,b^2\,a^2 + 4\,b^2\,a + 6\,a\,c + b^2\right.$$

$$+ 6\,c)\,t^3 + (70\,a^4 b^4 + 35\,a^3 b^4 + 15\,b^4 a^2 + 30\,a^2 b^2 c + 5\,a\,b^4 + 36\,c\,b^2 a + b^4 + 18\,b^2 c + c^2)\,t^4 + \mathrm{O}(t^5)$$

Comparison with the diagonal itself:

```
> S:=mtaylor(F,[x,y,z],3*N):
```

```
> diag:=add(coeff(coeff(coeff(S,x,i),y,i),z,i)*x^i,i=0..N);
```

$$diag := 1 + (2\,a + 1)\,b\,x + (6\,b^2 a^2 + 3\,b^2 a + b^2 + c)\,x^2 + b\,(20\,a^3 b^2 + 10\,b^2 a^2 + 4\,b^2 a + 6\,a\,c + b^2 + 6\,c)\,x^3 + (70\,a^4 b^4 + 35\,a^3 b^4 + 15\,b^4 a^2 + 30\,a^2 b^2 c + 5\,a\,b^4 + 36\,c\,b^2 a + b^4 + 18\,b^2 c + c^2)\,x^4$$

Transcendental Diagonals (sec. 5.2)

1/(1 -x -y -z^2)/(1-x-x*y)

```
> R:=1/(1 -x -y -z^2)/(1-x-x*y);
```

$$R := \frac{1}{(-z^2 - x - y + 1)\,(-x\,y - x + 1)}$$

From HolonomicFunctions:

```
> dop:=-(x^5*(-256+27*x^2)*(-16+3125*x^2)*(-10240000+1228395306880*
  x^2-946275\
  29933129*x^4-1153450428192365*x^6+165650655591225*
  x^8+1353715988625*x^1\
  0))*Dx^7-(x^4*(-838860800000+9082941916708864O*x^2
  -35699486716767279l0\
  4*x^4+2059155474565602861552*x^6+2368601270185682950218 5*x^8
  -697779898\
  1495512228125*x^10+4933017885019925688 75*x^12+39976925289082031 25*
  x^14\
  ))*Dx^6-6*x^3*(-865075200000+79949257252864000*x^2
  -54772201446602122 2\
  4*x^4+3281751359056793296136*x^6+3334097451884967419257 0*x^8
  -129414021\
  96250328222325*x^10+110574138399758154975 0*
  x^12+82382078632760156 25*x^1\
  4)*Dx^5-30*x^2*(-382730240000+26768867640279040*x^2
  -43763786604090587 6\
  48*x^4+2515335505474026452016*x^6+2179301246185533891176 3*x^8
  -13492716\
  168474984809820*x^10+130862271687133168717 5*
  x^12+89666762796548437 50*x\
  ^14)*Dx^4-15*x*(-524288000000+18588486652395520*x^2
  -14780673805091843 0\
  720*x^4+7448111749262291138592*x^6+4941847939384159723403 9*x^8
  -6873744\
  267777956492920 5*x^10+697973271994820968492 5*
```

```
x^12+437530497072680261 25\
*x^14)*Dx^3-45*(-20971520000-328475773501440*x^2
-2845024692769928192 0*x\
^4+11290833019611992838 40*x^6+29784904284569559556 83*x^8
-2632528143369\
2776481555*x^10+25653589724319647711 25*x^12+14514682262784078375*
x^14)*D\
x^2-315*x*(15156222361600-388216867194142720*
x^2+12234257394729005696*x\
^4-19928710815094882592 5*x^6-15299201708841068322 35*
x^8+13189731655474\
2455325*x^10+654525741648153375*x^12)*Dx-645120*
(163840000+64401425100\
80*x^2+392786350587920*x^4-9713021909301736*x^6-57338528994215525*
x^8+3\
997762795636575*x^10+16244591863500*x^12):
```

```
> map2(degree,dop,[Dx,x]);
```

$$[7, 19]$$

Order 7 and degree 19.

Check how many initial conditions are needed:

```
> deq:=DEtools[diffop2de](dop,[Dx,x],y(x)):
> gfun:-diffeqtorec(deq,y(x),u(n));
```

$$\begin{aligned}
&\big(-114[\ldots 12 \text{ digits}\ldots]375\, n^7 - 159[\ldots 13 \text{ digits}\ldots]250\, n^6 - 945[\ldots 13 \text{ digits}\ldots]500\, n^5 \\
&\quad - 305[\ldots 14 \text{ digits}\ldots]000\, n^4 - 583[\ldots 14 \text{ digits}\ldots]000\, n^3 - 656[\ldots 14 \text{ digits}\ldots]000\, n^2 \\
&\quad - 404[\ldots 14 \text{ digits}\ldots]000\, n - 104[\ldots 14 \text{ digits}\ldots]000\big)\, u(n) + \big(-128[\ldots 14 \text{ digits}\ldots]375\, n^7 \\
&\quad - 403[\ldots 15 \text{ digits}\ldots]250\, n^6 - 524[\ldots 16 \text{ digits}\ldots]500\, n^5 - 372[\ldots 17 \text{ digits}\ldots]000\, n^4 \\
&\quad - 156[\ldots 18 \text{ digits}\ldots]000\, n^3 - 391[\ldots 18 \text{ digits}\ldots]000\, n^2 - 539[\ldots 18 \text{ digits}\ldots]000\, n \\
&\quad - 316[\ldots 18 \text{ digits}\ldots]000\big)\, u(n+2) + \big(229[\ldots 15 \text{ digits}\ldots]075\, n^7 + 858[\ldots 16 \text{ digits}\ldots]650\, n^6 \\
&\quad + 142[\ldots 18 \text{ digits}\ldots]600\, n^5 + 134[\ldots 19 \text{ digits}\ldots]600\, n^4 + 788[\ldots 19 \text{ digits}\ldots]000\, n^3 \\
&\quad + 283[\ldots 20 \text{ digits}\ldots]000\, n^2 + 575[\ldots 20 \text{ digits}\ldots]200\, n + 506[\ldots 20 \text{ digits}\ldots]000\big)\, u(n+4) + \big(\\
&\quad -915[\ldots 15 \text{ digits}\ldots]905\, n^7 - 429[\ldots 17 \text{ digits}\ldots]190\, n^6 - 857[\ldots 18 \text{ digits}\ldots]680\, n^5 \\
&\quad - 947[\ldots 19 \text{ digits}\ldots]840\, n^4 - 624[\ldots 20 \text{ digits}\ldots]400\, n^3 - 245[\ldots 21 \text{ digits}\ldots]880\, n^2 \\
&\quad - 531[\ldots 21 \text{ digits}\ldots]800\, n - 489[\ldots 21 \text{ digits}\ldots]680\big)\, u(n+6) + \big(-711[\ldots 14 \text{ digits}\ldots]688\, n^7 \\
&\quad - 454[\ldots 16 \text{ digits}\ldots]632\, n^6 - 123[\ldots 18 \text{ digits}\ldots]600\, n^5 - 186[\ldots 19 \text{ digits}\ldots]800\, n^4 \\
&\quad - 168[\ldots 20 \text{ digits}\ldots]272\, n^3 - 905[\ldots 20 \text{ digits}\ldots]568\, n^2 - 270[\ldots 21 \text{ digits}\ldots]040\, n \\
&\quad - 343[\ldots 21 \text{ digits}\ldots]200\big)\, u(n+8) + \big(137[\ldots 13 \text{ digits}\ldots]544\, n^7 + 102[\ldots 15 \text{ digits}\ldots]760\, n^6 \\
&\quad + 332[\ldots 16 \text{ digits}\ldots]184\, n^5 + 600[\ldots 17 \text{ digits}\ldots]200\, n^4 + 653[\ldots 18 \text{ digits}\ldots]256\, n^3 \\
&\quad + 427[\ldots 19 \text{ digits}\ldots]200\, n^2 + 156[\ldots 20 \text{ digits}\ldots]016\, n + 244[\ldots 20 \text{ digits}\ldots]560\big)\, u(n+10) + \big(\\
&\quad -503[\ldots 10 \text{ digits}\ldots]480\, n^7 - 408[\ldots 12 \text{ digits}\ldots]880\, n^6 - 141[\ldots 14 \text{ digits}\ldots]240\, n^5 \\
&\quad - 272[\ldots 15 \text{ digits}\ldots]800\, n^4 - 313[\ldots 16 \text{ digits}\ldots]920\, n^3 - 216[\ldots 17 \text{ digits}\ldots]120\, n^2
\end{aligned}$$

$$- 830[\text{...17 digits...}]560\, n - 136[\text{...18 digits...}]000 \big) u(n+12) + \big(41943040000\, n^7$$
$$+ 4068474880000\, n^6 + 169[\text{...9 digits...}]000\, n^5 + 390[\text{...10 digits...}]000\, n^4$$
$$+ 539[\text{...11 digits...}]000\, n^3 + 448[\text{...12 digits...}]000\, n^2 + 206[\text{...13 digits...}]000\, n$$
$$+ 407[\text{...13 digits...}]000 \big) u(n+14)$$

```
> N:=15:
> S:=mtaylor(R,[x,y,z],3*N):
> S:=add(coeff(coeff(coeff(S,z,i),y,i),x,i)*x^i,i=0..N):
> Deq:={deq,seq((D@@i)(y)(0)=coeff(S,x,i)*i!,i=0..13)}:
> st:=time():gfun:-istranscendental(Deq,y(x)),time()-st;
```

true, "multiple root of multiplicity 2 of the indicial equation ==> ln at 0", 1.040

Note that dop is minimal, but not irreducible:

```
> DEtools[DFactor](dop,[Dx,x]);
```

$$\Big[-x^5 \left(27 x^2 - 256\right) \left(3125 x^2 - 16\right) \left(1353715988625\, x^{10} + 165[\text{...9 digits...}]225\, x^8\right.$$
$$\left. - 115[\text{...10 digits...}]365\, x^6 - 94627529933129\, x^4 + 1228395306880\, x^2 - 10240000\right) \Big(Dx^4$$
$$+ \big(2 \big(148[\text{...13 digits...}]875\, x^{14} + 185[\text{...15 digits...}]250\, x^{12} - 278[\text{...16 digits...}]725\, x^{10}$$
$$+ 100[\text{...17 digits...}]520\, x^8 + 886[\text{...15 digits...}]080\, x^6 - 151[\text{...14 digits...}]064\, x^4$$
$$+ 353[\text{...11 digits...}]360\, x^2 - 335544320000\big) Dx^3\big) \big/ \big(x \left(27 x^2 - 256\right) \big(3125 x^2$$
$$- 16\big) \big(1353715988625\, x^{10} + 165[\text{...9 digits...}]225\, x^8 - 115[\text{...10 digits...}]365\, x^6$$
$$- 94627529933129\, x^4 + 1228395306880\, x^2 - 10240000\big)\big) + \big(\big(246[\text{...14 digits...}]875\, x^{14}$$
$$+ 341[\text{...16 digits...}]875\, x^{12} - 443[\text{...17 digits...}]875\, x^{10} + 134[\text{...18 digits...}]625\, x^8$$
$$+ 135[\text{...17 digits...}]200\, x^6 - 227[\text{...15 digits...}]384\, x^4 + 268[\text{...12 digits...}]960\, x^2$$
$$- 3103784960000\big) Dx^2\big) \big/ \big(x^2 \left(27 x^2 - 256\right) \left(3125 x^2 - 16\right) \big(1353715988625\, x^{10}$$
$$+ 165[\text{...9 digits...}]225\, x^8 - 115[\text{...10 digits...}]365\, x^6 - 94627529933129\, x^4 + 1228395306880\, x^2$$
$$- 10240000\big)\big) + \big(3 \big(252[\text{...14 digits...}]375\, x^{14} + 385[\text{...16 digits...}]625\, x^{12}$$

$- 452[\ldots 17\text{ digits}\ldots]675\, x^{10} + 101[\ldots 18\text{ digits}\ldots]775\, x^{8} + 122[\ldots 17\text{ digits}\ldots]840\, x^{6}$

$- 222[\ldots 15\text{ digits}\ldots]352\, x^{4} + 940[\ldots 11\text{ digits}\ldots]680\, x^{2} - 1509949440000\,)\, Dx\,) \big/ (x^{3}\, (27\, x^{2}$

$- 256\,)\, (3125\, x^{2} - 16\,)\, (1353715988625\, x^{10} + 165[\ldots 9\text{ digits}\ldots]225\, x^{8}$

$- 115[\ldots 10\text{ digits}\ldots]365\, x^{6} - 94627529933129\, x^{4} + 1228395306880\, x^{2} - 10240000\,)\,)$

$+ \,(15\, (460[\ldots 13\text{ digits}\ldots]125\, x^{14} + 782[\ldots 15\text{ digits}\ldots]675\, x^{12} - 880[\ldots 16\text{ digits}\ldots]505\, x^{10} + 118[\ldots 17\text{ digits}..$

$- 16\,)\, (1353715988625\, x^{10} + 165[\ldots 9\text{ digits}\ldots]225\, x^{8} - 115[\ldots 10\text{ digits}\ldots]365\, x^{6}$

$- 94627529933129\, x^{4} + 1228395306880\, x^{2} - 10240000\,)\,)\,),\, Dx^{3} + \frac{(243\, x^{2} - 1024)\, Dx^{2}}{(27\, x^{2} - 256)\, x}$

$+ \frac{(501\, x^{2} - 448)\, Dx}{x^{2}\, (27\, x^{2} - 256)} + \frac{64\, (3\, x^{2} + 1)}{x^{3}\, (27\, x^{2} - 256)} \Big]$

A solution of its adjoint:

```
> adjdop:=DEtools[adjoint](dop,[Dx,x]):
> DEtools[formal_sol](adjdop,[Dx,x],x=0,order=6);
```

$\Big[-\frac{\ln(x)}{2} - \frac{5}{4} + \Big(-\frac{4481944159\, \ln(x)}{32000} - \frac{21121726441}{89600} \Big) x^{2} + \Big($

$-\frac{422423733107545481\, \ln(x)}{16000000} - \frac{522836458951349197353}{12544000000} \Big) x^{4} + \mathrm{O}(x^{6}),\, -\frac{1}{2}$

$- \frac{21121726441}{224000}\, x^{2} - \frac{278[\ldots 14\text{ digits}\ldots]991}{1568000000}\, x^{4} + \mathrm{O}(x^{6}),\, (-\ln(x) - 1)\, x + \Big($

$-\frac{2074230247\, \ln(x)}{9600} - \frac{25573195771}{144000} \Big) x^{3} + \Big(-\frac{718649082330303005837\, \ln(x)}{192000000}$

$- \frac{162431341372873711997}{576000000} \Big) x^{5} + \mathrm{O}(x^{6}),\, -x - \frac{2074230247}{9600}\, x^{3} - \frac{718[\ldots 13\text{ digits}\ldots]837}{192000000}\, x^{5}$

$+ \mathrm{O}(x^{6}),\, x^{2} + \frac{21121726441}{112000}\, x^{4} + \mathrm{O}(x^{6}),\, \sqrt{x}\, \Big(-1 - \frac{307340003093}{1280000}\, x^{2}$

$- \frac{254[\ldots 18\text{ digits}\ldots]071}{5898240000000}\, x^{4} + \mathrm{O}(x^{6}) \Big),\, \sqrt{x}\, \Big(x + \frac{358463112577}{1792000}\, x^{3}$

$+ \frac{968[\ldots 16\text{ digits}\ldots]297}{288358400000}\, x^{5} + \mathrm{O}(x^{6}) \Big) \Big]$

Take the second solution:

```
> S:=%[2];
```

$$S := -\frac{1}{2} - \frac{21121726441}{224000} x^2 - \frac{278[\ldots 14\ \text{digits}\ldots]991}{1568000000} x^4 + O(x^6)$$

```
> deq:={DEtools[diffop2de](adjdop,[Dx,x],y(x)),seq((D@@i)(y)(0)=coeff
  (S,x,i)*i!,i=0..4)}:
> istranscendental(deq,y(x));
```

true, "multiple root of multiplicity 2 of the indicial equation ==> ln at 0"

1/(1-x-y-z^2)/(1-x-x*y^2)

```
> R:=1/(1 -x -y -z^2)/(1-x-x*y^2);
```

$$R := \frac{1}{(-z^2 - x - y + 1)(-x y^2 - x + 1)}$$

Operator computed by HolonomicFunctions:

```
> dop:=DEtools[de2diffop](op(select(has,DeqZ9,x)),y(x),[Dx,x]):
> map2(degree,numer(dop),[Dx,x]);
```

[9, 60]

Order 9 and degree 60

```
> st:=time():istranscendental(DeqZ9,y(x));time()-st;
```

true, "multiple root of multiplicity 2 of the indicial equation ==> ln at 0"

10.058

Generalized Apéry sums (special cases of Prop. 13 in the Appendix)

```
> _MAXORDER:=100:
  Z:=proc(p,q)
  local n,k,j,i,Sn,rec,u,unk,S,ZZ;
      unk:=binomial(n,k)^p*binomial(n+k,n)^q;
      ZZ:=SumTools[Hypergeometric][Zeilberger](unk,n,k,Sn);
      if has(ZZ[1],n) then
            S:=op(1,ZZ);
         rec:={add(coeff(S,Sn,i)*u(n+i),i=0..degree(S,Sn)),
            seq(u(i)=add(eval(unk,[n=i,k=j]),j=0..i),i=0..degree(S,
  Sn)-1)}
      else rec:=gfun:-listtorec([seq(add(eval(unk,[n=i,k=j]),j=0..i),
  i=0..20)],u(n),['ogf'])[1]
      fi;
      gfun:-rectodiffeq(rec,u(n),y(z),'homogeneous'=true)
  end:
> for pq in [seq(seq([i,j],j=0..3),i=0..3)] do p,q:=op(pq);st[p,q]:=
  time();
     print(pq,istranscendental(Z(p,q),y(z)),time()-st[p,q]);st[p,q]:=
```

```
time()-st[p,q] od:
```

[0, 0], *FAIL*, 0.341

[0, 1], *FAIL*, 0.100

[0, 2], *true*, "multiple root of multiplicity 2 of the indicial equation ==> ln at infinity", 0.044

[0, 3], *true*, "multiple root of multiplicity 2 of the indicial equation ==> ln at RootOf(_Z^2+27)", 0.074

[1, 0], *FAIL*, 0.035

[1, 1], *FAIL*, 0.039

```
istranscendental: input diff eqn of order 3, minimal one of order 2
```

[1, 2], *true*, "multiple root of multiplicity 2 of the indicial equation ==> ln at 0", 0.152

```
istranscendental: input diff eqn of order 8, minimal one of order 4
```

[1, 3], *true*, "multiple root of multiplicity 3 of the indicial equation ==> ln at 0", 0.481

[2, 0], *FAIL*, 0.030

[2, 1], *true*, "multiple root of multiplicity 2 of the indicial equation ==> ln at 0", 0.055

[2, 2], *true*, "multiple root of multiplicity 3 of the indicial equation ==> ln at 0", 0.066

```
istranscendental: input diff eqn of order 12, minimal one of order 6
```

[2, 3], *true*, "multiple root of multiplicity 4 of the indicial equation ==> ln at 0", 1.427

[3, 0], *true*, "multiple root of multiplicity 2 of the indicial equation ==> ln at 0", 0.050

```
istranscendental: input diff eqn of order 8, minimal one of order 4
```

[3, 1], *true*, "multiple root of multiplicity 3 of the indicial equation ==> ln at 0", 0.830

```
istranscendental: input diff eqn of order 16, minimal one of order 6
```

[3, 2], *true*, "multiple root of multiplicity 4 of the indicial equation ==> ln at 0", 1.761

```
istranscendental: input diff eqn of order 25, minimal one of order 9
```

[3, 3], *true*, "multiple root of multiplicity 5 of the indicial equation ==> ln at 0", 9.239

The 3 cases that fail with p_0=1 are exactly those where the power series is proved to be algebraic in Prop. 13. Those with p_0=0 and p_1 in {0,1} are also easily seen to be algebraic.

The total time for this session is about 2 min.:

```
> time()-T0;
```

129.295